
• Wprowadzenie: klasteryzacja spektralna

• Tworzenie pakietu SAS/IML

• Tworzenie rozszerzenia do Enterprise Miner

• Prezentacja działania



Wstęp

• Rozwiązanie powstało w ramach pracy inżynierskiej związanej 
z klasteryzacją spektralną

• Pytanie: jak dostarczyć algorytm, by jego użycie było jak 
najprostsze?

• Odpowiedź: IML Packages + Enterprise Miner Extension Node



Wprowadzenie: Klasteryzacja spektralna

• Rodzaj klasteryzacji opierający się na spektralnej teorii grafów

• Zyskał popularność dzięki solidnym podstawom 
matematycznym, prostej implementacji i konkurencyjnym 
wynikom

• Szeroko używany w wielu obszarach, takich jak statystyka, 
uczenie maszynowe, rozpoznawanie wzorców, eksploracja 
danych i analiza obrazów



IML Packages



SAS IML



SAS IML

• Język i narzędzie do operacji na 
macierzach, zapewniający zestaw 
funkcji z zakresu statystyki, analizy i 
wizualizacji danych

• Przypomina składnią języki takie jak 
Matlab i R, pozwala na integrację z R

• Może być wykonywany z poziomu 
4GL procedurą proc iml;



IML 
Packages

Mechanizm dodany stosunkowo 
niedawno, bo w wersji 14.1 (2016r.)

Pozwala na dzielenie się kodem, 
wraz z dokumentacją, programami i 
danymi za pomocą 
ustrukturyzowanego archiwum zip

Wykorzystuje komendę 
packages pozwalającą na 
instalowanie i wczytywanie 
zawartości pakietu



Komenda 
packages



package install '{ścieżka do pliku zip}';

• instaluje pakiet umieszczając 
zawartość archiwum w 
odpowiedniej lokalizacji. Zależnie 
od środowiska, w którym 
wywołana jest komenda pakiet 
może zostać zainstalowany dla 
wszystkich lub dla danego 
użytkownika



package load {nazwa pakietu};

• wczytuje pakiet, wykonując 
kod zawarty w plikach źródłowych. 
Najczęściej są to definicje funkcji, 
do których otrzymujemy dostęp 
po wykonaniu polecenia



package help {nazwa pakietu};

• wyświetla w konsoli dokumentację 
pakietu w postaci tekstowej*



package list;

• wyświetla wszystkie dostępne 
pakiety



package uninstall {nazwa pakietu};

• odinstalowuje pakiet usuwając 
odpowiednie pliki



package info {nazwa pakietu};

• wyświetla metadane pakietu 
(krótki opis, 
wersja, lokalizacja instalacji itd.)



package libname {nazwa biblioteki} 
{nazwa pakietu};

• tworzy bibliotekę 
zawierającą dane dostarczane z 
pakietem

• Wymaga załadowania pakietu 
poleceniem load



Tworzenie pakietu



Struktura folderu

• Data – dane wczytywane
poleceniem package libname

• Help - dokumentacja

• Programs - przykładowe programy
lub dodatkowe narzędzia

• Source – kod pakietu uruchamiany
poleceniem package load

• Info.txt - metadane



data

• Dane wczytywane poleceniem 
package libname

• Mogą to być na dane 
wykorzystywane w przykładach

• Pliki w formacie tabel SAS



help

• Dokumentacja w formacie txt 
wyświetlana 
poleceniem package help

• Można dodatkowo umieścić tu 
bardziej kompletną dokumentację 
w formacie pdf lub html

• Pliki muszą mieć nazwę zgodną z 
nazwą pakietu



programs

• Programy dostarczane z pakietem 
(inne niż sam kod pakietu)

• Nie ma wymagań co do ich 
formatu, nie są wykorzystywane 
przez mechanizm pakietów

• Mogą to być przykładowe 
programy wykorzystujące nasz 
pakiet bądź dodatkowe narzędzia



source

• Właściwy kod pakietu w postaci 
plików IML

• Najczęściej są to definicje funkcji

• Zaleca się umieszczanie funkcji w 
osobnych plikach o nazwach 
zgodnych z nazwą funkcji

• Dobrą praktyką jest rozpoczynanie 
nazwy funkcji od prefiksu 
identyfikującego pakiet



Info.txt

• Zawiera metadane wyświetlane 
poleceniem package info

• Dodatkowo zawiera listę plików 
źródłowych - tylko pliki tu 
wymienione zostaną wykonane w 
momencie ładowania pakietu



Tworzenie rozszerzenia 
do SAS Enterprise Miner



SAS Enterprise 
Miner



SAS Enterprise Miner

• Graficzne narzędzie do 
modelowania i eksploracji 
danych

• Pozwala na szybkie testowanie 
wielu ścieżek przetwarzania 
danych i porównywanie 
wyników

• Zapewnia przejrzystość 
wykonywanych operacji



Extension 
Node

• Mechanizm pozwalający na 
rozszerzenie funkcjonalności 
programu przez użytkownika

• Może zostać w pełni 
zintegrowany z programem i 
łączyć się z innymi węzłami w 
diagramie



Extension 
Node –
potrzebne 
pliki

Katalog SAS z plikami z kodem 
SAS-owym zawierającymi 
program do wykonania

Plik XML definiujący właściwości 
węzła

Pliki GIF – ikony węzła



main.sas

• Zawiera główne makro %main

• Używa polecenia %include do wczytania zawartości innych 
plików

• Uruchamia pozostałe pliki z zależności od stanu węzła w 
diagramie

• Do rozpoznawania stanu węzła używana jest makrozmienna 
&EM_ACTION



%macro main;
%if %upcase(&EM_ACTION) = CREATE %then %do;

filename temp catalog 'sashelp.specc.spcl_create.source';

%include temp;

filename temp;

%create;

%end;

%else

%if %upcase(&EM_ACTION) = TRAIN %then %do;

filename temp catalog 'sashelp.specc.spcl_train.source';

%include temp;

filename temp;

%train;
%end;

%if %upcase(&EM_ACTION) = SCORE %then %do;

filename temp catalog 'sashelp.specc.spcl_score.source';

%include temp;

filename temp;

%score;

%end;

%if %upcase(&EM_ACTION) = REPORT %then %do;

filename temp catalog 'sashelp.specc.spcl_report.source';

%include temp;

filename temp;

%report;

%end;

%mend main;

%main;



create.sas

• Zawiera makro %create

• Uruchamiany, gdy węzeł zostaje dodany do diagramu

• Inicjalizacja właściwości/parametrów (%EM_PROPERTY)

• Rejestracja zbiorów/plików/katalogów (%EM_REGISTER)



%macro create;

%EM_PROPERTY(NAME=ClusterNum, VALUE=2);

%EM_PROPERTY(NAME=Neighborhood, VALUE=complete);

%EM_PROPERTY(NAME=Laplacian, VALUE=normalizedRW);

%EM_PROPERTY(NAME=NeighFun, VALUE=gaussian);

%EM_PROPERTY(NAME=Sigma, VALUE=1.0);

%EM_PROPERTY(NAME=K, VALUE=5);

%EM_PROPERTY(NAME=MaxIter, VALUE=10);

%EM_PROPERTY(NAME=Distance, VALUE=EUCLIDEAN);

%EM_PROPERTY(NAME=Seed, VALUE=12346);

%EM_PROPERTY(NAME=Impute, VALUE=none);

%EM_PROPERTY(NAME=Standard, VALUE=N);

%EM_PROPERTY(NAME=ClusRole, VALUE=SEGMENT, ACTION=SCORE);

%EM_REGISTER(key=import_data, type=DATA);

%EM_REGISTER(key=vectors, type=DATA);

%EM_REGISTER(key=indvectors, type=DATA);

%EM_REGISTER(key=OUTSTAT, type=DATA); 

%EM_REGISTER(key=graph_table, type=DATA); 

%EM_REGISTER(key=MODELINFO, type=DATA); 

%EM_REGISTER(key=CLUSTERSUM, type=DATA); 

%EM_REGISTER(key=ITERSTAT, type=DATA); 

%EM_REGISTER(key=OVERALLVARSTAT, type=DATA); 

%EM_REGISTER(key=CLUSTERBASEDVARSTAT, type=DATA);

%mend create;



train.sas

• Odpowiada za przetworzenie danych wejściowych i 
przeprowadzenie klasteryzacji spektralnej

• Zawiera główne makro %train oraz pomocnicze makro 
%getNObs

• Wczytuje zarejestrowane wcześniej zbiory makrem 
%EM_GETNAME



Makro pomocnicze

%macro getNObs(inds, nobs);

/*macro for setting macrovariable nobs with number of observations in 

dataset inds*/

%global &nobs; 

data _null_; 

set &inds end=eof; 

if eof then call symputx("&nobs", _N_); 

run; 

quit; 

%mend getNObs;



A co gdyby zrobić to sprytniej?

%macro getNObs(inds, nobs);

%global &nobs;                          

data _null_;                            

if 0 then set &inds nobs=n;           

call symputx("&nobs", n);

stop; 

run;                                  

quit;                              

%mend getNObs;



Walidacje – mechanizm wyjątków

• Sprawdzenie czy istnieją dane

%if (^%sysfunc(exist(&EM_IMPORT_DATA)) and
^%sysfunc(exist(&EM_IMPORT_DATA, VIEW)))
or "&EM_IMPORT_DATA" eq "" %then %do;
%let EMEXCEPTIONSTRING = exception.server.IMPORT.NOTRAIN,1;
%goto doenda;

%end;

<reszta kodu makra>

%doenda:
%mend train;



Walidacje

• Sprawdzenie czy istnieje chociaż jedna zmienna przedziałowa, 
porządkowa lub binarna

%if (%EM_INTERVAL_INPUT %EM_ORDINAL_INPUT %EM_BINARY_INPUT eq ) %then %do;

%let EMEXCEPTIONSTRING = ERROR;

%put &em_codebar;

%put Error: Must use at least one interval, ordinal or binary input;

%put &em_codebar;

%goto doenda;

%end;



Użycie pakietu IML

proc iml;

package load spectralclust;

varnames = { %EM_INTERVAL_INPUT %EM_ORDINAL_INPUT %EM_BINARY_INPUT };

print 'Used variables:';

print varnames;

use &em_user_import_data;

read all var varnames into m;

close &em_user_import_data;

m = spccSpectralize(m, &EM_PROPERTY_ClusterNum , "&EM_PROPERTY_Laplacian.",

"&EM_PROPERTY_NeighFun.", &EM_PROPERTY_Sigma , 

"&EM_PROPERTY_Neighborhood.", &EM_PROPERTY_K );

create &em_user_vectors from m;

append from m;

close &em_user_vectors;

CALL symput("nvar", putn(ncol(m), "BEST6."));

quit;



Szybka klasteryzacja - HPCLUS

• Używa algorytmu k-średnich do klasteryzacji

• Korzysta z wielu wątków procesora

• Tworzy wiele zbiorów zawierających statystyki klasteryzacji



ods listing exclude Standardization ;                              

filename flowtemp "&em_file_emflowscorecode";                   

ods output PerformanceInfo = _tmp_outperformanceinfo

ModelInfo = &em_user_modelinfo

NObs = _tmp_outnobs

ClusterSum = &em_user_clustersum

IterStats = &em_user_ITERSTAT

DescStats = &em_user_OVERALLVARSTAT

WithinClusStats = &em_user_CLUSTERBASEDVARSTAT

Timing = _tmp_outtiming; 

proc hpclus data = &em_user_vectors 

maxclusters = &EM_PROPERTY_ClusterNum 

maxiter = &EM_PROPERTY_MaxIter

outstat = &EM_USER_OUTSTAT  

distance = &EM_PROPERTY_Distance

Seed = &EM_PROPERTY_Seed;                   

input %DO i=1 %to &nvar; COL&i %END;;             

score out=_tmp_out_score;

code file=flowtemp;                               

&hpdm_performance.;                               

run;



Wyjściowe zbiory danych

data &em_export_train;

set &em_user_import_data;

set _tmp_out_score(drop=_DISTANCE_);

run;

data &em_user_vectors;

set &em_user_vectors;

set _tmp_out_score;

run;



score.sas

• Zawiera makro %score odpowiadające za zmianę 
metadanych zbioru wyjściowego

• Uruchamiany po zakończeniu makra %train



Zmiana metadanych

%macro score;

filename _FHPTS  "&EM_FILE_CDELTA_TRAIN";         

data _null_;                                         

file _FHPTS;                                     

put 'if upcase(NAME) eq "_DISTANCE_" then Role="REJECTED";';   

put "if upcase(NAME) eq '_CLUSTER_ID_' then do;";              

put "   Role='&em_property_clusrole';";                        

put "   Level='NOMINAL';";                                     

put "end;";                                                    

run;                                                               

filename _FHPTS;

%mend score;



report.sas

• Zawiera makro %report odpowiadające za generowanie 
raportów, które wyświetlają się następnie w oknie Rezultaty

• Uruchamiany po zakończeniu makra %score

• Do tworzenia wizualizacji używane jest makro %EM_REPORT



Tworzenie 
raportów

%em_report(key=indvectors,
viewtype=Lattice,
latticetype=Lineplot,
block=eigenvectors,
views=3,
autodisplay=Y,
x=i,
y=value,
latticex=variable,
equalizerowy=Y,
description=Eigenvector Profiles);



Tworzenie 
raportów

%em_report( key=vectors,

viewtype=Data,

block=eigenvectors,

views=1,

autodisplay=Y,

description=Eigenvector Table);



Tworzenie raportów

%EM_REPORT(key=CLUSTERSUM, 
viewtype=PIE, 

x=Cluster, 

freq=Frequency, 

block=PLOT, 

autodisplay=Y,

description=segmentplotlabel ) 



Tworzenie katalogu

%macro CreateSources(dirPath, libraryname, catname);

libname &libraryname "&dirPath";
filename src_cat catalog "&libraryname..&catname..spcl.source";
filename mydata "&dirPath.\main.sas";
data _null_;

file src_cat;
infile mydata;
input;
put _infile_;

run;
<wczytanie reszty plików>

proc catalog cat=&libraryname..&catname.;
copy out=sashelp.&catname.;

run;
%mend CreateSources;



Tworzenie katalogu

%macro createCatalog(userPath);

%CreateSources(&userPath\SpectralClusteringSAS\SAS EM\SpecClust, 

specc, specc)

LIBNAME specc CLEAR;

%mend createCatalog;



Tworzenie katalogu - wywołanie

%include "Path_to_repository\SpectralClusteringSAS\SAS 

EM\SpecClust\create_catalog.sas";

%createCatalog(Path_to_repository);



Właściwości węzła

• Daje użytkownikowi możliwość 
podania hiperparametrów do 
algorytmu

• Umożliwia ograniczenie  
wartości możliwych do podania

• Wygląd okna zdefiniowany jest 
w pliku XML



Okno dialogowe



Przykładowa definicja hiperparametru



Układ hiperparametrów



Prezentacja działania





Przydatne linki

• Writing Packages: A New Way to Distribute and Use SAS/IML® Programs
https://support.sas.com/resources/papers/proceedings16/SAS4201-2016.pdf

• SAS/IML User's Guide - Packages
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.3/imlug/imlug_pack
ages_sect001.htm

• SAS Enterprise Miner Extension Nodes: Developer's guide
https://documentation.sas.com/doc/pl/emxndg/15.1/titlepage.htm

• Procedura HPCLUS  
https://documentation.sas.com/doc/en/emhpprcref/14.2/emhpprcref_hpclus_
overview01.htm

• Kod naszego rozwiązania: https://github.com/Pastuszka/SpectralClusteringSAS

https://support.sas.com/resources/papers/proceedings16/SAS4201-2016.pdf
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.3/imlug/imlug_packages_sect001.htm
https://documentation.sas.com/doc/pl/emxndg/15.1/titlepage.htm
https://documentation.sas.com/doc/en/emhpprcref/14.2/emhpprcref_hpclus_overview01.htm
https://github.com/Pastuszka/SpectralClusteringSAS


Dziękujemy za 
uwagę


